

The Effect of Bethanechol on Tracheobronchomalacia in Infants

Chary Akmyradov Ph.D., Arkansas Children's Research Institute;

ABSTRACT

This single-center retrospective cohort study included 33 subjects (11 cases, 22 controls) hospitalized between 2017 and 2022. Cases treated with Bethanechol were matched with controls who did not receive Bethanechol. Case-control matching of 1:2 was performed based on birth gestational age, sex, the severity of Tracheobronchomalacia (TBM), the severity of bronchopulmonary dysplasia, and respiratory support. The primary outcome was the effect of Bethanechol on the mean Pulmonary Severity Score. This presentation demonstrates the statistical methods, visualization, and SAS® programming behind this clinical study.

INTRODUCTION


TBM is characterized by airway collapse resulting in increased work of breathing leading to prolonged mechanical ventilation, increased caloric needs, and prolonged hospitalization. Bethanechol, a muscarinic agonist, has demonstrated improvement in trachealis tone in animal models, but no trials have studied efficacy in infants with TBM. A pulmonary severity score (PSS) has been developed as an assessment tool to compare infant lung disease. A lower PSS reflects improved respiratory status. Cases and controls were observed for 105 days. Each case was started the Bethanechol treatment on different days. First two cases started the medicine at day 28, and all cases were on this treatment by day 70.

CHALLENGES

One of the challenges in this repeated observations is each patient started the treatment at a different time point. Additionally, newborn infants are developing quickly, this may show up as a trend on the outcome variable for controls.

VISUAL DATA EXPLORATION

First, the daily mean PSS were plotted against days by grouping the cases and controls (Figure 2) using long data (Figure 1), where id is the subjects from 1 to 33, case=1 for cases and 0 for controls, day is the observation day from 1 to 105. The drawback of this graph is case PSS means from day 28 to 70 is calculated using both cases who started and who did not start the medicine. However, after day 70, the line graph shows some separation between controls and cases while daily standard deviations have become smaller for cases.



	id	case	PSS	day
2309	32	0	0.8	105
2310	33	0	0.42	105
2311	1	1	0.95	1
2312	2	1	1	1
2313	3	1	1.15	1

Figure 1. A portion of the long data which is used to produce Figure 2.




Figure 2. Daily Mean PSS were decreased after Bethanechol treatment when compared with controls.

SAS Code for Figure 2 using long data:

```

/*obtaining means and standard deviations */
proc means data = pulm_score_long;
  by case day;
  var pss;
  output out = meansout mean=mean std = std;
run;
/*creating lower and upper bounds*/
data reshape(drop=stderr);
  set meansout;
  lower=mean - std;
  upper=mean + std;
run;
ods graphics / attrpriority=none;
proc sgplot data=reshape;
  styleattrs datasymbols = (circlefilled trianglefilled)
             datacontrastcolors= (blue red);
  scatter x=day y=mean / yerrorlower = lower
                      yerrorupper = upper
                      group = case;
  series x=day y=mean / group = case ;

```

```

axis          values = (0 to 110 by 10)
              label   = 'Days from 40 Week Gestational Age';
yaxis        values = (0 to 2 by 0.25)
              label   = 'Mean +/- St.Dev. of PSS';
refline 70 /axis = x label='All cases on Bethanechol: day 70'
              lineattrs = (color=black pattern=dash );
refline 28 / axis = x label='First two cases on Bethanechol: day 28'
              lineattrs = (color=black pattern=dash );
format case case.; label case = 'Group';
run;

```

Is there any association between a case's pre-treatment mean PSS, post-treatment mean PSS, and the day of starting the treatment? To answer this question I created a gradient colored scatter plot using 11 cases' pre and post treatment mean PSS in the Figure 3. If the observation is below diagonal line then the variable at the x-axis is higher than the y-axis and vice versa. If the observations are closer to the diagonal line then pre and post treatment mean PSS are closer to each other. Observations with blue tones have started the treatment earlier than the red tones. In Figure 3, two blue observations are clearly away from the diagonal line indicates earlier treatment is more effective:




Figure 3. Observations below the diagonal line indicate higher pre-treatment mean PSS than post-treatment mean PSS.

This is the SAS code for Figure 3:

```

proc sgplot data = pss_means;
  scatter x = mean_pss_pre y = mean_pss_post
          / colorresponse = pss_beth_start_day
            transparency = 0.08
            markerattrs = (symbol=circlefilled size=15);
  lineparm x = 0 y = 0 slope = 1 /

```

```

        legendlabel = 'Pre & Post PSS Equality';
    xaxis label = 'Mean PSS at Pre-Bethanechol Treatment';
    yaxis label = 'Mean PSS at Post-Bethanechol Treatment';
    label pss_beth_start_day='Bethanechol Start Day';
run;

```

Figure 2 indicates the cases' and controls' progress over the 105 days of observations but not everybody started the treatment on the same day. This brings the following question: What if we align all of the cases and matched controls with respect to their treatment beginning days? For this task, the treatment start days are subtracted from the original days in the long data (Figure 1). This also enables us to see clear difference between pre and post treatment outcomes. Figure 4 indicates that the cases' daily mean PSS dropped after the treatment. One may ask the reason for both groups getting closer PSS after 80 days. The top portion indicates the number of observations (cases and controls) in the given day in the data. After day 70 there is only 2 observations and 4 controls which is not showing a significant difference in mean PSS:




Figure 4. Treatment start day aligned matched case-control daily mean PSS comparison.

The SAS code for Figure 4:

```

proc sort data = pulm_score_long_strata;
by case Day_from_Beth_Start;
run;

```

```

proc means data = pulm_score_long_strata;
  by case Day_from_Beth_Start;
  var pss;
  output out = meansout mean=mean std = std;
run;
data reshape;
  set meansout;
  lower=mean - std;
  upper=mean + std;
run;
ods graphics /attrpriority=none;
proc sgplot data=reshape;
  styleattrs   datasymbols      = (circlefilled trianglefilled)
              datacontrastcolors= (blue red);
  scatter      x = Day_from_Beth_Start y=mean / yerrorlower   = lower
              yerrorupper   = upper
              group          = case;
  series       x = Day_from_Beth_Start y = mean /group        = case;
  series       x = Day_from_Beth_Start y = _freq_ / y2axis
              group          = case;
  y2axis       offsetmin      = 0.8
              values         = (0 to 20 by 5) grid
              label          = "Control vs Case Count"
              labelpos       = top;
  xaxis        values         = (-70 to 80 by 10)
              label          = 'Days from Bethanechol Treatment';
  yaxis        values         = (0.25 to 1.75 by 0.25)
              offsetmax     = 0.2
              label          = 'Mean +/- St.Dev. of PSS';
  refline 0    / axis         = x
              lineattrs     = (color=black pattern=dash );
  format case case.;   label case = 'Group';
run;

```

DATA ANALYSIS WITH PROC MIXED

The primary goal of this study was to investigate the effect of Bethanechol on TBM. Using mixed models I compared percent change of pre and post treatment PSS values among matched cases and controls:

```

proc mixed data=pss.pss_means_strata method=ml;
  class id strata case;
  model diff_pct= case / ddfm=kr solution;
  random intercept/ subject=strata type=cs;
  lsmeans case / pdiff cl;
run;

```

In the above code, strata is the matching indicator, diff_pct is percent change between pre and post mean PSS values.

As a result cases' mean and 95%CI of mean PSS change is 20.84% (2.1%, 39.58%) lower than the controls' with two sided t-test p-value of 0.0309.

CONCLUSION

The goal of this paper is to demonstrate how to utilize mixed linear modeling and how to support the study findings using SGPLOT procedures. If the outcome is a binary variable then one can apply conditional logistic regression. If the study is 1 to 1 matching, then paired t-test or rank test can be used depending on the normality test.

REFERENCES

Tips and Tricks for Clinical Graphs using ODS Graphics, Sanjay Matange
(<https://support.sas.com/resources/papers/proceedings11/281-2011.pdf>)

SAS Institute Inc. Base SAS® 9.4 Procedures Guide. Cary, NC: SAS Institute Inc.

SAS Institute Inc. SAS/STAT® 9.4 User's Guide. Cary, NC: SAS Institute Inc.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Chary Akmyradov, Ph.D.
Arkansas Children's Research Institute
13 Children's Way,
Slot 842,
Little Rock, Arkansas, 72202
(501)364-2107
AkmyradovC@archildrens.org
<https://www.akmyradovc.com>

TRADEMARK

SAS® and all other SAS® Institute Inc. product or service names are registered trademarks or trademarks of SAS® Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies